
Úvod do Operačních Systémů

6. cvičení
Zpracování textu programem awk.

Obsah
● awk = Aho–Weinberger–Kernighan

● vzory

● akce

● pole

● funkce

● rozšíření awk (nawk)

awk – vzory I
● Výpis prvních 10 řádků (head)

cat -n /etc/passwd | awk 'NR<=10'

cat -n /etc/passwd | awk 'NR==1,NR==10'

● Výpis posledních 10 řádků (tail)
awk NR\>=$((`wc -l </etc/passwd`-10)) /etc/passwd

● Výpis řádků obsahujících RE (grep)
awk /root/ /etc/group

● Výpis řádků neobsahujících RE (grep -v)
awk '!/root/' /etc/group

awk – vzory II
● Výběr položek z ypcat passwd podle jmen

ypcat passwd \
| awk '/Adela/ /Barbora/ /Celeste/ /Dana/ /Eliska/'

ypcat passwd \
| awk -f /home/courses/Y36UOS/common/cv06/names.awk

● Z výstupu ypcat passwd výběr položek s UID 1000-9999

ypcat passwd | awk -F: '$3>=1000 && $3<=9999'

● Zjištění počtu uživatelů z výstupu ypcat passwd

ypcat passwd \
| awk 'END { print "Total users: " NR }'

awk – akce I
● Výběr a změna pořadí jednotlivých položek /etc/passwd

awk 'BEGIN { FS=":"; OFS=":" } 
          { print $3,$1,$5 }' /etc/passwd

● Nalezení uživatele s nejdelším jménem

ypcat passwd | awk -F: '
{ 

if (length($5) > max) {
max=length($5); user=$1; name=$5

}
}
END {

print user ":" name " (" max ")" 
}'

awk – akce II

● Formátovaný výpis uživatelů Korn shellu

ypcat passwd | awk -F: '
BEGIN {

print "Uzivatele Korn shellu:"
for (d=0; d<=60; d++) printf "-"; printf "\n"

}
$7 ~ /ksh/ {

if (length($1)<=7) tab="\t\t"; else tab="\t"
print $1 tab $5
ksh++

}
END {

for (d=0; d<=60; d++) printf "-"; printf "\n"
print "Celkem " ksh " z " NR " uzivatelu"

}'

nawk – pole I
● Výpis uživatelů ve skupině x36alg

ypcat group | nawk -F: '
/^x36alg/ {

items=split($4,a,",")
for (i=1; i<=items; i++) print a[i]

}'

nawk – pole II +
● Frekvenční slovník

man man | awk '
BEGIN { FS="[^a-zA-Z]" }

{ for (i=1;i<=NF;i++) words[tolower($i)]++ }

END {
# prázdné položky nepočítáme
delete words[""]
for (i in words) {

sum++
print words[i],i

}
print "V textu bylo " sum " r zných slov"ů

}' | sort -n

awk – uživatelské funkce +
ls / | nawk '

function swap(x,y) {
tmp=item[x]
item[x]=item[y]
item[y]=tmp

}

{ item[NR]=$0 }

END {
srand();
for (i=1; i<=NR; i++) {

RAND=int(NR*rand())+1
swap(i,RAND)

}
for (i=1; i<=NR; i++) { print item[i] }

}'

● Náhodné pořadí vstupních řádků

Příprava na příští cvičení
● Vyjděte z minulé přípravy...

● Ke každému řádku souboru seznam.csv přidejte položku 
obsahující uživatelské jméno uvedeného studenta.

● Vytvořte HTML stránku obsahující tabulku se všemi údaji 
obsaženými v souboru seznam.csv (pokud možno 
obecně, nezávisle na počtu řádků i položek v souboru).


